

多重比較法の検討

2024年10月18日 第10回大阪SAS勉強会

免責事項

本発表で述べる意見や見解は、発表者個人のものであり、
所属する組織や企業の公式な立場や意見を代表するものではありません。



目的

- Williamsの方法, Dunnettの方法, 固定順序法はどれも薬効薬理試験にて用いられる検定
 - 薬効薬理試験では, 低用量・中用量・高用量等複数用量を比較, 検定するため, 多重性の問題を考慮し, 第一種の過誤確率を5%に保つ必要がある.
- ☆ どの手法を用いるべきか, 各手法の内容を確認しSASで実行する.



背景知識

検定の多重性問題：検定を繰り返し行うことで、全体としての結論の第1種の過誤確率が大きくなってしまうこと。

例えば、有意水準5%で独立に5回検定を行うと、少なくとも1回有意となる確率は
 $1 - 0.95^5 = 0.0975$

このように、有意水準5%で独立にn回検定を行うと、以下の確率で少なくとも有意となってしまう。
 $1 - 0.95^n$

1回1回の検定は有意水準5%で行われていても、全体としての有意水準が5%よりずっと大きくなってしまう。

そこで本発表では、多重性への対処法の1つである多重比較法を用いて、全体としての有意水準を5%に制御することを考える。



本発表で扱う多重比較法

本発表では代表的なパラメトリック法であるDunnettの方法とWilliamsの方法について取り扱う。
(パラメトリック法にはTukeyの方法等ほかの方法も存在する)

- **Dunnettの方法：**

1つの対照群と2つ以上の処理群があつて、母平均について対照群と処理群の対比較のみを同時に検定するための多重比較法。

- **Williamsの方法：**

1つの対照群と2つ以上の処理群があつて、各群に順序付けがあるものとする。母平均に単調性（増加・減少）を想定することができる時、母平均について対照群と処理群の対比較のみを検定するための多重比較法。



Dunnettの方法_手順

事前に帰無仮説，対立仮説，有意水準を設定しておく。

各処理群の母平均が対照群の母平均と比べ，「異なるか」だけでなく，「小さいといえるかどうか（大きいほうに関心はない）」，「大きいといえるかどうか（小さいほうに関心はない）」の3通りの状況が考えられる。

それぞれの帰無仮説に対する対立仮説として3通り考えられる。

A) $\mu_1 \neq \mu_i$

B) $\mu_1 > \mu_i$

C) $\mu_1 < \mu_i$

上記を踏まえて手法を確認する。



Dunnettの方法_手順

①各群の平均 \bar{x}_i および分散 V_i を計算する.

②誤差自由度 ϕ_E および誤差分散 V_E を計算する

$$\phi_E = N - a = n_1 + n_2 + \dots + n_a - a$$

$$V_E = \frac{\sum_{i=1}^a (n_i - 1) V_i}{\phi_E}$$

③すべての $i = 2, 3, \dots, a$ に対して統計検定量 t_{1i} を計算する

$$t_{1i} = \frac{\bar{x}_1 - \bar{x}_i}{\sqrt{V_E \left(\frac{1}{n_1} + \frac{1}{n_i} \right)}} \quad (i = 2, 3, \dots, a)$$



Dunnettの方法_手順

④次式により ρ を求める。

$$\rho = \frac{n_2}{n_2 + n_1}$$

⑤ Dunnettの方法のための両側（上側）5%点の表を参考に $d(\alpha, \phi_E, \rho; \alpha)$ を求める。

Aの場合 : $|t_{1i}| \geq d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を棄却し μ_1 と μ_i には差があると判断する。

$|t_{1i}| < d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を保留する。

Bの場合 : $t_{1i} \geq d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を棄却し μ_1 と μ_i には差があると判断する。

$t_{1i} < d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を保留する。

Cの場合 : $t_{1i} \leq -d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を棄却し μ_1 と μ_i には差があると判断する。

$t_{1i} > -d(\alpha, \phi_E, \rho; \alpha)$ ならば帰無仮説を保留する。



Williamsの方法_手順

事前に帰無仮説，対立仮説，有意水準を設定しておく。

Dunnettの方法の設定に加えて，さらにa個の母平均について以下の関係を想定できるものとする。

$$\mu_1 \geq \mu_2 \geq \dots \geq \mu_i$$

$$\mu_1 \leq \mu_2 \leq \dots \leq \mu_i$$

上記を踏まえて手法を確認する。

①データをランダムサンプリングし，それぞれの群ごとの平均 \bar{x}_i および分散 V_i を計算する。

②誤差自由度 ϕ_E および誤差分散 V_E を計算する

$$\phi_E = N - a = n_1 + n_2 + \dots + n_a - a$$

$$V_E = \frac{\sum_{i=1}^a (n_i - 1) V_i}{\phi_E}$$



Williamsの方法_手順

③ $p = a$ とおく.

④ T_i (第*i*群のデータの合計)を用いて, 次の統計量 (第*p*群以下の隣接した処理群を含めた平均値) を計算する.

$$y_{2p} = \frac{T_2 + T_3 + \cdots + T_p}{n_2 + n_3 + \cdots + n_p}$$

$$y_{3p} = \frac{T_3 + \cdots + T_p}{n_3 + \cdots + n_p}$$

$$y_{pp} = \frac{T_p}{n_p}$$



Williamsの方法_手順

⑤ $y_{2p}, y_{3p}, \dots, y_{pp}$ の最大値 M_p を求める。

$$M_p = \max\{y_{2p}, y_{3p}, \dots, y_{pp}\}$$

⑥ 統計検定量 t_p を計算する。

$$t_p = \frac{M_p - \bar{x}_1}{\sqrt{V_E \left(\frac{1}{n_p} + \frac{1}{n_1} \right)}}$$



Williamsの方法_手順

⑦Williamsの方法のための両側（上側）5%点の表を参考に $w(p, \phi_E; \alpha)$ を求める。

単調増加の場合、

$t_p < w(p, \phi_E; \alpha)$ なら帰無仮説を保留して検定手順を終了する。

$t_p \geq w(p, \phi_E; \alpha)$ となるなら帰無仮説を棄却して「 μ_p は μ_1 より大きい」と判断し次の手順へ進む。

⑧ $p = 2$ なら手順を終了する。 $p \geq 3$ のときは p の値を1だけ減らしてこれを新たに p として手順④から繰り返す。



SASでの実行

使用したデータ：統計的多重比較法の基礎より引用[1]

DRUGID	用量	ni	RESPONCE						
1	0 mg	7	415	380	391	413	372	359	401
2	15 mg	7	387	378	359	391	362	351	348
3	30 mg	7	357	379	401	412	392	356	366
4	60 mg	7	361	351	378	332	318	344	315
5	90 mg	7	299	308	323	351	311	285	297



SASによる実行 Dunnettの方法

```
proc format;  
    value $DRUG '1' = '0mg' '2' = '15mg' '3' = '30mg' '4' =  
    '60mg' '5' = '90mg';  
run;  
  
proc glm data = Pra_data order=data;  
    class DRUGID;  
    model RESPONCE = DRUGID;  
    lsmeans DRUGID/adjust=dunnett pdiff=control('0mg');  
    format DRUGID $DRUG.;  
run;
```



SAS結果 Dunnettの方法

SAS システム

GLM プロシジャ
最小 2 乗平均
多重比較の調整: Dunnett

DRUGID	RESPONCE の最小 2 乗平均	H0:LSMean=Control
		Pr > t
0mg	390.142857	
15mg	368.000000	0.1747
30mg	380.428571	0.8017
60mg	342.714286	0.0008
90mg	310.571429	<.0001

- 60mg, 90mgで優位差が見られた。



SASによる実行 Williamsの方法 手順1

SAS など多くの市販されている統計ソフトでは Williams 検定が標準では実施できない。

⇒マクロを作成し対応。（新版 実用SAS 生物統計ハンドブック[2]より）

```
/*Williams*/
```

```
%macro WILP;
```

```
proc glm data = &DATA outstat=VAR noprint;  
    class &GROUP;  
    model &Y=&GROUP / ss1;  
run;
```

Proc glmの実行

```
data VAR; set VAR;  
    keep VAR DF;  
    VAR=SS/DF;  
    if _TYPE_='ERROR' then output;  
run;
```

誤差分散を算出

```
proc sort data=&DATA; by &GROUP;  
run;
```



SASによる実行 Williamsの方法 手順2

```
proc summary data=&DATA;  
    var &Y;  
    by &GROUP;  
    output out=OUT mean=MEAN n=N;  
run;
```

応答値について、薬剤の用量ごとに要約統計量を計算

```
proc transpose data=OUT out=N prefix=N;  
    var N;  
    id &GROUP;  
run;
```

```
proc transpose data=OUT out=MEAN prefix=M;  
    var MEAN;  
    id &GROUP;  
run;
```



SASによる実行 Williamsの方法 手順3-1

```
data SUMMARY; merge N MEAN VAR;
run;

data SUMMARY;set SUMMARY;
drop S1-S100 NN1-NN100;
retain SIN 0;
array NN(&G) N1-N&G;
array M(&G) M1-M&G;
array MM(&G) MM1-MM&G;

MM1=M1;
array S(10, 10) S1-S100;
array NNN(10,10) NN1-NN100;

do I=1 to &G-1;

do J=1 to &G-1;
s(I, J) =0;
NNN(I, J) =0;
do K=I to J;
S(I, J) =S(I, J) +M(K+1) *NN(K+1) ;
NNN(I, J) =NNN(I, J) +NN(K+1);
end;
if NNN(I, J) ^=0 then S(I, J)=S(I, J) /NNN(I, J);
end;
end;
```

NN:サンプルサイズを格納
M:各グループの平均値を格納
MM:調整平均値を格納

1つ目のグループの平均値をMM1
に格納
グループ間平均の合計が格納され
ているS配列とグループサイズの
合計を格納するNNN配列を定義



SASによる実行 Williamsの方法 手順3-2

```
data SUMMARY; merge N MEAN VAR;
  run;

data SUMMARY;set SUMMARY;
  drop S1-S100 NN1-NN100;
  retain SIN 0;
  array NN(&G) N1-N&G;
  array M(&G) M1-M&G;
  array MM(&G) MM1-MM&G;

  MM1=M1;
  array S(10, 10) S1-S100;
  array NNN(10,10) NN1-NN100;
```

組み合わせごとにグループ間の
平均値とサンプルサイズを計算

```
do I=1 to &G-1;

  do J=1 to &G-1;
    S(I, J) =0;
    NNN(I, J) =0;
    do K=I to J;
      S(I, J) =S(I, J) +M(K+1) *NN(K+1) ;
      NNN(I, J) =NNN(I, J) +NN(K+1);
    end;
    if NNN(I, J) ^=0 then S(I, J)=S(I, J) /NNN(I, J);
  end;
end;
```



SASによる実行 Williamsの方法 手順4-1

調整平均値の計算
MM(J+1)に各グループの平均値を格納
TAIL が1の場合,最大値をとり,
TAIL が2の場合, 最小値をとる

```
do J=1 to &G-1;
    S(I, J));
    S(I, J));
    MM(J+1) =M(J+1);
    do I=1 to &G-1;
        if NNN(I, J) ^=0 and 1=&TAIL then MM(J+1)=max(MM(J+1),
        if NNN(I, J) ^=0 and 2=&TAIL then MM(J+1)=min(MM(j+1),
    end;
end;
```

CONTROL:最初のグループ間の
調整平均値
その後, グループの調整平均値
と平均値を格納

```
do I=&G to 2 by -1;
    CONTROL = MM1;
    GROUP=I;
    ADJMEAN=MM(I);
    MEAN=M(I);
    T=(MM(i) -MM1) / (VAR*(1/NN(i) + 1/N1))**0.5;
    if 2=&TAIL then T=T*(-1);
    WP=1-probmc('williams' , T, ., DF, I-1);
    WTC=probmc('williams', ., 1-&A, DF, I-1);
    STAR=' ';
    if WP<=&A then SIN=1;
    if SIN=1 then STAR='*';
    output;
    SIN=0;
end;
```

run;



SASによる実行 Williamsの方法 手順4-2

```
do J=1 to &G-1;
    MM(J+1) =M(J+1);
    do I=1 to &G-1;
        if NNN(I, J) ^=0 and 1=&TAIL then MM(J+1)=max(MM(J+1),
S(I, J));
        if NNN(I, J) ^=0 and 2=&TAIL then MM(J+1)=min(MM(j+1),
S(I, J));
    end;
end;
```

```
do I=&G to 2 by -1;
    CONTROL = MM1;
    GROUP=I;
    ADJMEAN=MM(I);
    MEAN=M(I);
    T=(MM(i) -MM1) / (VAR*(1/NN(i) + 1/N1))**0.5;
    if 2=&TAIL then T=T*(-1);
    WP=1-probmc('williams', T, ., DF, I-1);
    WTC=probmc('williams', ., 1-&A, DF, I-1);
    STAR=' ';
    if WP<=&A then SIN=1;
    if SIN=1 then STAR='*';
    output;
    SIN=0;
end;
run;
```

T: t_p の計算
WP: Williamsのp値を計算
(前述の手順ではWilliamsの方法のための
両側(上側)5%点の表を参考)
WTC: $w(p, \phi_E; \alpha)$ を求める



SASによる実行 Williamsの方法

```
proc sort; by GROUP; run;

proc print;
    var CONTROL GROUP MEAN ADJMEAN T WP WTC STAR;
run;

%mend;

%let DATA=Pra_data;
%let G=5;
%let A=0.05;
%let GROUP=DRUGID;
%let Y=RESPONCE;
%let TAIL=2;

%WILP;
```



SAS結果 Williamsの方法

SAS システム

Obs	CONTROL	GROUP	MEAN	ADJMEAN	T	WP	WTC	STAR
1	390.143	2	368.000	368.000	1.96909	0.029120	1.69726	*
2	390.143	3	380.429	374.214	1.41648	0.098610	1.77581	
3	390.143	4	342.714	342.714	4.21767	0.000112	1.80143	*
4	390.143	5	310.571	310.571	7.07604	0.000000	1.81381	*

- 順序の逆転が見られた箇所はMEANとADJMEANが異なる.
- 90 mg群, 60 mg群で優位差が見られ30 mg群では $T < WTC$ となるので帰無仮説を保留して検定作業を終了する.



まとめ

- Dunnettの方法とWilliamsの方法はどちらも薬効薬理試験にてよく用いられる。
 - 単調性を満たす場合、Williamsのほうが検出力が高いと言われている。
⇒例数設計の際、サンプルサイズを小さく抑えられる可能性がある。
 - 各群のサンプル数が異なる場合は対処する必要がある。
(Williamsの方法はすべてのサンプルサイズが等しいことを前提としているが、欠測の度合いによっては、欠測が発生しても使用可能な事例もある)
- ☆ **試験の特徴（動物の脱落が起こりやすい等）や手法の特徴を考慮し、適切な手法を事前に選択することが重要。**



参考資料

[1] 永田 靖, 吉田 道弘, 統計的多重比較法の基礎, サイエンティスト社 (1997)

[2] 浜田 知久馬, 新版 実用SAS 生物統計ハンドブック (2017)

[3] 第二期医薬安全性研究会 20170610第20回定例会公開資料 浜田final

[4] 中西 展大, 橋本 敏夫, 浜田知久馬 (2014) , 薬理薬効試験におけるこれからの標準的検定手法 薬理薬効試験における Williams 多重比較検定の妥当性と有用性
総説 日薬理誌 144, 185~191

